

```

/*
    Clarity_ETL checks for today's Clarity token drop.

    If the token dropped then:
        Check if this the first time we've seen the token drop for this particular job (jobid)?
        Yes: abort the job
        No: Record a success drop in the trigger table for the job.  e-mail success.
    If the token did not drop then abort the job.

    If the current time is greater than the check time then send out a failure e-mail
*/

%macro Clarity_ETL(jobid=,checktime=);

%let etllib = /apps/sas/datasets/data26/NCREVCYC/data/Clarity_ETL;
libname etllib "&etllib";

/* Test if Clarity token dropped */
options nomprint;
proc sql;
connect to teradata as hcclnc(db=hcclnc tdpid=tdp1 user=&CLR_NC_USERID
password=&CLR_NC_PW);
create table Clarity_ETL (compress=yes) as
select * from connection to hcclnc
(
Select
    TokenDate,
    TokenTime
from publicinfo.TokenXDailyETL
where TokenDate = current_date and TokenStatus = 'Drop'
);
disconnect from hcclnc;
quit;
options mprint;

/* Create macro variable (observations) that contains observations in Clarity_ETL */
proc sql;
    select count(*) into :observations
        from Clarity_ETL;
quit;

%put &observations;

/* Set abort return value if no Clarity ETL */
%if &observations eq 0 %then %do;

/* If the current time is past the parameter time, then send out failure message */
%let time      = %sysfunc(time(),time8.0);
%let checktimeehhmmss = &checktime.:00;
%if %sysvalf("&time">"&checktimeehhmmss"&t) %then %do;
filename failure email
    to="finance.and.reporting@kp.org"
/*      to="stephen.x.fiorelli@kp.org"      */
    from="finance.and.reporting@kp.org"
    subject="(Auto e-mail) SAS BI job(&jobid.) failed to trigger from Clarity ETL"

```

```

;

data _null_;
    file failure;
run;
%end;

%abort return 99;

%end;

/* Clarity ETL occurred if we get here      */

/* Check if we have already set the Clarity ETL flag */
/* If this is the first time then e-mail success      */

%let status = N;

PROC SQL;
    select
        clarity_service_status into :status
        from etllib.clarity_job_trgr
        where clarity_service_trigger_date = today() and jobid=&jobid;
QUIT;

%put status = &status;

/*
    If we already set the status flag to ETL complete, then we
    don't have to do it again.  Simply abort so we don't trigger
    (already triggered once.
*/

%if &status=Y %then %abort return 99;

filename success email
    to="finance.and.reporting@kp.org"
/*    to="stephen.x.fiorelli@kp.org"      */
    from="finance.and.reporting@kp.org"
    subject="(Auto e-mail) SAS BI job(&jobid.) successfully triggered from Clarity ETL"
;

data _null_;
    file success;
run;

PROC SQL noprint;
    INSERT INTO etllib.clarity_job_trgr
        SET jobid = &jobid,
            clarity_service_status = "Y",
            clarity_service_trigger_date = today();
QUIT;

%mend;

```